Newton Divided Differences and Interpolation Polynomials

Mth 351 May 8 2001 - Maple 5.00
Bent E. Petersen
Filename: 351s2001_newton_divided_diff.mws

```
> restart;
```

The procedure divdiff() computes symbolic Newton divided differences recursively. Not much error checking is done so be careful. Note the use of a limit in the code. This allows us to compute the divided differences even when the nodes are not all distinct. The purpose of this procedure is to give you something to experiment with when you study Newton divided differences.

We use the divdiff() procedure to compute interpolation polynomials of functions in the procedure Ninterp() below. One nice feature is that Ninterp() does the right thing for the case of repeated nodes. Of course, Maple does have built-in interpolation, and one should use it. The Maple routine however works with lists of points, rather than functions. Thus Maple has no sensible way of handling repeated nodes, and returns an error in this case. We show below how the Maple routine can be used to interpolate functions and how to handle repeated nodes in that case simply by blowing up the repeated points and then passing to a limit.

```
> divdiff:=proc(f)
>  local x;
>  if nargs < 2 then ERROR(FAIL); fi;
>  if nargs = 2 then RETURN(f(args[2]));
> else
> 
  limit((divdiff(f,args[3..nargs])-divdiff(f,x,args[3..nargs-1]))/(a
  rgs[nargs]-x),x=args[2]);
> fi;
> end:
```

Here's some examples:

```
> divdiff(f,a);
```

$$f(a)$$

```
> divdiff(f,a,a);
```

$$D(f)(a)$$

```
> divdiff(f,a,b);
```

$$-\frac{f(b)-f(a)}{-b+a}$$

```
> divdiff(f,a,a,a);
```

$$\frac{1}{2}(D^{(2)})(f)(a)$$

```
> divdiff(f,a,a,b);
```

$$\frac{f(b) - f(a) - D(f)(a)\,b + D(f)(a)\,a}{(-b+a)^2}$$

```
> divdiff(f,a,b,c);
```

$$-\frac{f(c)\,b - f(c)\,a + f(b)\,a - f(b)\,c + f(a)\,c - f(a)\,b}{(-c+b)\,(-b+a)\,(a-c)}$$

Here's a routine to compute interpolation polynomials using Newton's divided differences formula. Note Maple does have built in support for interpolation. One should really use it. See below.

```
> Ninterp:=proc(f,x)
> local k,A,n;  n:=3;
> if nargs < n then ERROR(FAIL); fi;
> if nargs = n then RETURN(x->f(args[3]));
> else
> A:=divdiff(f,args[n..nargs]);
> for k from nargs-1 to n by -1 do
  A:=A*(x-args[k])+divdiff(f,args[n..k]); od;
> fi;
> A;
> end:
```

Here's a few examples.

The tangent line

```
> Ninterp(f,t,a,a);
```

$$D(f)(a)\,(t-a) + f(a)$$

The parabola of contact order 2, that is, the Taylor polynomial at a of degree at most 2:

```
> Ninterp(f,t,a,a,a);
```

$$\left(\frac{1}{2}(D^{(2)})(f)(a)\,(t-a) + D(f)(a)\right)(t-a) + f(a)$$

The parabola with order of contact 1 at a and order of contact 0 at b

```
> Ninterp(f,s,a,a,b);
```

$$\left(\frac{(\mathrm{f}(b) - \mathrm{f}(a) - \mathrm{D}(f)(a)\, b + \mathrm{D}(f)(a)\, a)\,(s - a)}{(-b + a)^2} + \mathrm{D}(f)(a)\right)(s - a) + \mathrm{f}(a)$$

The polynomial of degree at most 3 with order of contact 1 at a and at b:

```
> p:=Ninterp(f,x,a,a,b,b);
```

$$p := \left(\left(\frac{(-\mathrm{D}(f)(b)\, b + \mathrm{D}(f)(b)\, a + 2\,\mathrm{f}(b) - 2\,\mathrm{f}(a) - \mathrm{D}(f)(a)\, b + \mathrm{D}(f)(a)\, a)\,(x - b)}{(-b + a)^3} + \frac{\mathrm{f}(b) - \mathrm{f}(a) - \mathrm{D}(f)(a)\, b + \mathrm{D}(f)(a)\, a}{(-b + a)^2}\right)(x - a) + \mathrm{D}(f)(a)\right)(x - a) + \mathrm{f}(a)$$

Let's check this last polynomial:

```
> pf:=unapply(p,x):
> pf(a);
```

$$\mathrm{f}(a)$$

```
> pf(b): simplify(%);
```

$$\mathrm{f}(b)$$

```
> D(pf)(a);
```

$$\mathrm{D}(f)(a)$$

```
> D(pf)(b): simplify(%);
```

$$\mathrm{D}(f)(b)$$

Using Maple's interpolation facility

Let's now try to obtain this last polynomial by using Maple's built-in interpolation. The Maple routine requires that the nodes all be distinct, so first we use distinct nodes, and then we pass to the limit as two of the nodes approach the other two nodes. Note the use of the map() function to obtain a list of ordinates from the list of abscissas.

```
> p2:=interp([a,aa,b,bb],map(f,[a,aa,b,bb]),x):
> p3:=limit(limit(p2,aa=a),bb=b);
```

$$\begin{aligned} p3 := (&a^2\, b\, x\, \mathrm{D}(f)(b) - \mathrm{D}(f)(a)\, b^2\, a^2 + 3\,\mathrm{f}(a)\, a\, b^2 + a^2\,\mathrm{D}(f)(b)\, b^2 - 3\, a\, x^2\,\mathrm{f}(b) + 3\, a\, x^2\,\mathrm{f}(a) \\ &- x^3\,\mathrm{D}(f)(b)\, b + x^2\,\mathrm{D}(f)(b)\, b^2 + \mathrm{D}(f)(a)\, b^3\, a - a^3\,\mathrm{D}(f)(b)\, b + a^3\, x\,\mathrm{D}(f)(b) - a^2\, x^2\,\mathrm{D}(f)(a) \\ &- 2\, a\, x\,\mathrm{D}(f)(b)\, b^2 - a\, x\,\mathrm{D}(f)(a)\, b^2 + a\, x^2\,\mathrm{D}(f)(b)\, b - 3\,\mathrm{f}(b)\, a^2\, b + \mathrm{f}(b)\, a^3 - \mathrm{f}(a)\, b^3 \\ &- 2\, x^3\,\mathrm{f}(a) + 2\, x^3\,\mathrm{f}(b) - x^3\,\mathrm{D}(f)(a)\, b + x^3\,\mathrm{D}(f)(a)\, a - x\,\mathrm{D}(f)(a)\, b^3 + 3\, x^2\,\mathrm{f}(a)\, b - 3\, x^2\,\mathrm{f}(b)\, b \\ &- 2\, x^2\,\mathrm{D}(f)(b)\, a^2 + 6\, x\,\mathrm{f}(b)\, a\, b - 6\, x\,\mathrm{f}(a)\, a\, b + 2\, x\,\mathrm{D}(f)(a)\, a^2\, b - x^2\,\mathrm{D}(f)(a)\, a\, b \\ &+ x^3\,\mathrm{D}(f)(b)\, a + 2\, x^2\,\mathrm{D}(f)(a)\, b^2) \,\big/\, (-b + a)^3 \end{aligned}$$

It's not immediately obvious that p3 is equal to p, so let's look at the difference.

```
> simplify(p-p3);
```

$$0$$

Sure enough!

```
>
```